```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

Let us consider the Dirac equations for a massless particle defined by the matrix

```
> R:=matrix(4,4,[d[4],0,-i*d[3],-(i*d[1]+d[2]),0,d[4],-i*d[1]+d[2],i*d[3],
> i*d[3],i*d[1]+d[2],-d[4],0,i*d[1]-d[2],-i*d[3],0,-d[4]]);
```

$$R := \begin{bmatrix} d_4 & 0 & -i\,d_3 & -i\,d_1 - d_2 \\ 0 & d_4 & -i\,d_1 + d_2 & i\,d_3 \\ i\,d_3 & i\,d_1 + d_2 & -d_4 & 0 \\ i\,d_1 - d_2 & -i\,d_3 & 0 & -d_4 \end{bmatrix}$$

with entries in the ring $A = \mathbb{Q}(i)[d_1, d_2, d_3, d_4]$ of differential operators in $d_1, \ldots, d_4$ with coefficients in $\mathbb{Q}(i)$:

```
> A:=DefineOreAlgebra(diff=[d[1],x[1]],diff=[d[2],x[2]],diff=[d[3],x[3]],
> diff=[d[4],x[4]],polynom=[x[1],x[2],x[3],x[4]],comm=[i],
> alg_relations=[i^2+1]):
```

See, e.g., R. Courant, D. Hilbert, *Methods of Mathematical Physics*, Wiley Classics Library, Wiley, 1989. Let us consider the A-module $M = A^{1\times 4}/(A^{1\times 4}\, R)$ finitely presented by the matrix R and let us compute its endomorphism ring $E = \mathrm{end}_A(M)$:

```
> Endo:=MorphismsConstCoeff(R,R,A):
```

The A-module structure of the ring E can be generated by

```
> nops(Endo[1]);
```

$$18$$

generators which satisfy

```
> rowdim(Endo[2]);
```

$$22$$

A-linear relations. Let us try to compute idempotents of E defined by matrices over $\mathbb{Q}(i)$:

```
> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,0);
```

$$Idem := [[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1/2 & 0 & -1/2 & 0 \\ 0 & 1/2 & 0 & -1/2 \\ -1/2 & 0 & 1/2 & 0 \\ 0 & -1/2 & 0 & 1/2 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix},$$
$$\begin{bmatrix} 1/2 & 0 & 1/2 & 0 \\ 0 & 1/2 & 0 & 1/2 \\ 1/2 & 0 & 1/2 & 0 \\ 0 & 1/2 & 0 & 1/2 \end{bmatrix}],$$

$$[Ore\_algebra, [\text{“}diff\text{''}, \text{“}diff\text{''}, \text{“}diff\text{''}, \text{“}diff\text{''}], [x_1, x_2, x_3, x_4], [d_1, d_2, d_3, d_4], [x_1, x_2, x_3, x_4], [i], 0,$$
$$[], [i^2 + 1], [x_1, x_2, x_3, x_4], [], [], [diff = [d_1, x_1], diff = [d_2, x_2], diff = [d_3, x_3], diff = [d_4, x_4]]]]$$

We obtain the trivial idempotents 0 and id_M of E as well as two non-trivial idempotents e_1 and e_2 respectively defined by the matrices $Idem[1,2]$ and $Idem[1,4]$. Let us denote by $P = Idem[1,2]$ and $Q \in A^{4\times 4}$ defined by $R\,P = Q\,R$:

```
> P:=Idem[1,2]; Q:=Factorize(Mult(R,P,A),R,A);
```

$$P := \begin{bmatrix} 1/2 & 0 & -1/2 & 0 \\ 0 & 1/2 & 0 & -1/2 \\ -1/2 & 0 & 1/2 & 0 \\ 0 & -1/2 & 0 & 1/2 \end{bmatrix} \quad Q := \begin{bmatrix} 1/2 & 0 & 1/2 & 0 \\ 0 & 1/2 & 0 & 1/2 \\ 1/2 & 0 & 1/2 & 0 \\ 0 & 1/2 & 0 & 1/2 \end{bmatrix}$$

As the entries of the matrices P and Q belong to the field $\mathbb{Q}$ and $P^2 = P$ and $Q^2 = Q$, using linear algebraic techniques, we can easily compute bases of the free $\mathbb{Q}$-modules $\ker_{\mathbb{Q}}(.P)$, $\mathrm{im}_{\mathbb{Q}}(.P) = \ker_{\mathbb{Q}}(.(I_4 - P))$, $\ker_{\mathbb{Q}}(.Q)$ and $\mathrm{im}_{\mathbb{Q}}(.Q) = \ker_{\mathbb{Q}}(.(I_4 - Q))$ as follows:

```
> U1:=SyzygyModule(P,A): U2:=SyzygyModule(evalm(1-P),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q,A): V2:=SyzygyModule(evalm(1-Q),A):
> V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} -1 & 0 & -1 & 0 \\ 0 & -1 & 0 & -1 \\ 1 & 0 & -1 & 0 \\ 0 & -1 & 0 & 1 \end{bmatrix} \quad V := \begin{bmatrix} -1 & 0 & 1 & 0 \\ 0 & 1 & 0 & -1 \\ 1 & 0 & 1 & 0 \\ 0 & -1 & 0 & -1 \end{bmatrix}$$

In particular, the previous matrices define bases of the free A-modules $\ker_A(.P)$, $\mathrm{im}_A(.P)$, $\ker_A(.Q)$ and $\mathrm{im}_A(.Q)$. Hence, the unimodular matrices U and V, i.e., $U \in \mathrm{GL}_4(A)$ and $V \in \mathrm{GL}_4(A)$, are such that the matrices $U\,P\,U^{-1}$ and $V\,Q\,V^{-1}$ are block-diagonal formed by the diagonal matrices 0_2 and I_2:

```
> VERIF1:=Mult(U,P,LeftInverse(U,A),A);
> VERIF2:=Mult(V,Q,LeftInverse(V,A),A);
```

$$\mathit{VERIF1} := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \mathit{VERIF2} := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Then, the matrix R is equivalent to the block-diagonal matrix $S = V\,R\,U^{-1}$ defined by:

```
> S:=Mult(V,R,LeftInverse(U,A),A);
```

$$S := \begin{bmatrix} -i\,d_3 + d_4 & -i\,d_1 - d_2 & 0 & 0 \\ i\,d_1 - d_2 & -d_4 - i\,d_3 & 0 & 0 \\ 0 & 0 & d_4 + i\,d_3 & -i\,d_1 - d_2 \\ 0 & 0 & -i\,d_1 + d_2 & -i\,d_3 + d_4 \end{bmatrix}$$

This result can be directly obtained by using the command *HeuristicDecomposition*:

```
> HeuristicDecomposition(R,P,A)[1];
```

$$\begin{bmatrix} -i\,d_3 + d_4 & -i\,d_1 - d_2 & 0 & 0 \\ -i\,d_1 + d_2 & d_4 + i\,d_3 & 0 & 0 \\ 0 & 0 & d_4 + i\,d_3 & i\,d_1 + d_2 \\ 0 & 0 & -i\,d_1 + d_2 & -d_4 + i\,d_3 \end{bmatrix}$$

As we have $\mathrm{coim}_A(.P) \cong \mathrm{im}_A(.P)$ and $\mathrm{coim}_A(.Q) \cong \mathrm{im}_A(.Q)$, we obtain that the A-modules $\mathrm{coim}_A(.P)$ and $\mathrm{coim}_A(.Q)$ are free. Hence, by Theorem 4.2 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381, we obtain that R is equivalent to a block-triangular matrix. It can be obtained by computing bases of the free A-modules $\ker_A(.P)$, $\mathrm{coim}_A(.P)$, $\ker_A(.Q)$ and $\mathrm{coim}_A(.Q)$ as follows:

```
> Y2:=LeftInverse(Exti(Involution(Y1,A),A,1)[3],A): Y:=stackmatrix(U1,Y2);
> Z2:=LeftInverse(Exti(Involution(Z1,A),A,1)[3],A): Z:=stackmatrix(V1,Z2);
```

$$Y := \begin{bmatrix} -1 & 0 & -1 & 0 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix} \quad Z := \begin{bmatrix} -1 & 0 & 1 & 0 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

The matrices $Y \in \mathrm{GL}_4(A)$ and $Z \in \mathrm{GL}_4(A)$, respectively formed by the bases of $\ker_A(.P)$ and $\mathrm{coim}_A(.P)$ and by the bases of $\ker_A(.Q)$ and $\mathrm{coim}_A(.Q)$, are such that $T = Z\,R\,Y^{-1}$ is a block-triangular matrix defined by:

```
> T:=Mult(Z,R,LeftInverse(Y,A),A);
```

$$T := \begin{bmatrix} d_4 - i\,d_3 & i\,d_1 + d_2 & 0 & 0 \\ i\,d_1 - d_2 & d_4 + i\,d_3 & 0 & 0 \\ i\,d_3 & -i\,d_1 - d_2 & d_4 + i\,d_3 & -i\,d_1 - d_2 \\ -i\,d_1 + d_2 & -i\,d_3 & -i\,d_1 + d_2 & d_4 - i\,d_3 \end{bmatrix}$$

This last result can be directly obtained as follows:

```
> HeuristicReduction(R,P,A)[1];
```

$$\begin{bmatrix} d_4 - i\,d_3 & i\,d_1 + d_2 & 0 & 0 \\ i\,d_1 - d_2 & d_4 + i\,d_3 & 0 & 0 \\ i\,d_3 & -i\,d_1 - d_2 & d_4 + i\,d_3 & -i\,d_1 - d_2 \\ -i\,d_1 + d_2 & -i\,d_3 & -i\,d_1 + d_2 & d_4 - i\,d_3 \end{bmatrix}$$